DIAGRAMMI DI BODE

La risposta in frequenza di un sistema può essere vista in più modi diversi. Adesso studiamo il diagramma di Bode.

La risposta in frequenza è una rappresentazione della risposta del sistema ad un ingresso sinusoidale al variare della frequenza. Il teorema della risposta in frequenza dice che: “L'uscita di un sistema lineare ad un ingresso sinusoidale è una sinusoide con la stessa frequenza, con ampiezza uguale al prodotto dell'ampiezza della sinusoide di ingresso per il modulo della f.d.T. del sistema, e con la fase uguale alla somma della fase iniziale più la fase della f.d.T. del sistema.”

Noi useremo la risposta in frequenza di un sistema ad anello aperto per prevedere il suo comportamento ad anello chiuso e così intervenire per migliorare e/o garantire la stabilità.

Per mettere su grafico la risposta in frequenza, si deve creare un vettore di frequenze (da 0 ad infinito e quindi solo valori positivi) e calcolare il valore della funzione di trasferimento del processo a quelle frequenze. Supponiamo che G(s) sia la f.d.T. ad anello aperto e ω è il vettore delle frequenze, si disegna $G(j\omega)$ su ω. Ma siccome $G(j\omega)$ è un funzione complessa, è necessario disegnare sia il modulo sia la fase (diagrammi di Bode), o didegnare il vettore sul piano complesso di Gauss (diagrammi di Nyquist).

Per vedere i diagrammi di Bode di una f.d.T. si deve usare in Scilab la funzione ***bode***. Per esempio,

```
-->s=poly(0,'s');

-->h=syslin('c',(s^2+2*0.9*10*s+100)/(s^2+2*0.3*10.1*s+102.01))
 h  =
```

$$\frac{100 + 18s + s^2}{102.01 + 6.06s + s^2}$$

```
-->title='(s^2+2*0.9*10*s+100)/(s^2+2*0.3*10.1*s+102.01)';//titolo dei diagrammi di Bode

-->bode(h,0.01,100,title);
```

Notiamo la facilità con cui si determina la f.d.T. **h=syslin('c',(s^2+2*0.9*10*s+100)/(s^2+2*0.3*10.1*s+102.01))**



Notiamo che la frequenza sulle ascisse è su scala logaritmica, la fase è espressa in gradi e il modulo è dato in decibel. Ricordiamo che un dB è definito come: $20 * \log_{10}(|G(j\omega)|)$.

Margine di Guadagno e di Fase

Il *margine di guadagno* è definito come la variazione di guadagno ad anello aperto richiesta per rendere instabile il sistema. Sistemi che hanno ampi margini di guadagno possono sopportare maggiori variazioni dei parametri prima di diventare instabili ad anello chiuso.

Il *margine di fase* è definito come il cambio di fase ad anello aperto richiesto per rendere un sistema ad anello chiuso instabile.

Per i diagrammi di Bode, il margine di fase è definito come la differenza in fase tra la curva della fase e la retta a -180° nel punto corrispondente alla frequenza in cui la curva dell'ampiezza passa da 0 dB. Analogamente, il margine di guadagno è dato dalla differenza tra la curva del modulo e la curva a 0 dB nel punto in cui la fase vale -180°.

Esempio:

-->s=poly(0,'s');

-->N=50;

```
-->D=s^3+9*s^2+30*s+40;

-->ss=syslin('c',N,D);

-->bode(ss)
```



E' possibile calcolare i valori con Scilab usando le funzioni ***g_margin*** e ***p_margin.***

```
-->[g,fr]=g_margin(ss) //fr è la frequenza corrispondente e g è il margine di guadagno in dB.
 fr  =
    0.8717275
 g  =
    13.255157
```

//Dalla definizione si vede che il valore è negativo quindi il guadagno sarà minore di 1.

```
-->[p,fr]=p_margin(h)
 fr  =
    0.2941819
 p  =
   -79.33797
```

//In realtà è dato da 180 + p_margin

```
-->180+p_margin(h)

 ans  =

   100.66203
```

Se si varia il guadagno di un sistema il diagramma di Bode della fase non si modifica, a differenza del modulo.

Per esempio, supponiamo di aggiungere un guadagno di 100 in modo da avere al numeratore 5000.

Utilizzando i comandi del secondo metodo si ottiene:

```
-->s=poly(0,'s');
-->N=5000;
-->D=s^3+9*s^2+30*s+40;
-->h=N/D
 h  =
          5000
    ----------------
    40 + 30s + 9s² + s³
-->ss=syslin('c',h);
-->bode(ss)
```



Il grafico della fase non è cambiato, mentre il modulo è variato di 40 dB (guadagno relativo a 100).

Il margine di fase, ad occhio, è diventato negativo e quindi il sistema è ora instabile.

Anche adesso, si vuole calcolare i due margini con Scilab:

```
-->[p,fr]=p_margin(h)
 fr  =
   2.6885418
 p  =
   120.3744

-->[g,fr]=g_margin(h)
 fr  =
   0.8717275
 g  =
    -26.744843
```

La frequenza del margine di guadagno è rimasta la stessa , mentre il guadagno è variato. Ricordiamoci che è dato in dB ed è cambiato di segno.

Per il margine di fase si ha circa -59.625604°.

Banda passante

La banda passante è definita come l'intervallo di frequenza in cui l'attenuazione del modulo della risposta ad anello chiuso è inferiore a -3 dB. Tuttavia, essendo interessati a predire il comportamento del ciclo chiuso dalla risposta del sistema a ciclo aperto, useremo una approssimazione del secondo ordine del sistema, e, quindi, la banda passante è la frequenza alla quale il modulo a ciclo aperto è compresa tra -6 e -7.5 dB, supponendo la fase compresa tra -135° e 225°.

Mostreremo l'importanza della banda passante e di come l'uscita cambia con il variare della frequenza di ingresso. Vedremo che gli ingressi sinusoidali con frequenza minore della banda passante (B_P) sono seguiti abbastanza bene dal sistema. Invece, gli ingressi sinusoidali con frequenze maggiori di B_P sono attenuati (in modulo) di un fattore 0.707 o maggiore e sono anche variate in fase.

Esempio:

```
-->h=syslin('c',1/(s^2+0.5*s+1))
 h  =
        1
   --------------
   1 + 0.5s + s^2
-->bode(h,0.01,1) //inserisco anche la frequenza minima e massima.
```



Il diagramma mostra che la B_P è di circa 1.4 rad/sec (corrisponde sul diagramma a 0.22 Hz). Il diagramma mostra anche che per una frequenza di circa 0.05 Hz (0.3 rad/sec) l'uscita sinusoidale ha modulo quasi uguale all'ingresso e con poco sfasamento. Invece, per una frequenza di ingresso di 3 rad/sec (0.5 Hz), il modulo dell'uscita è circa -20 dB (o 1/10 dell'ingresso) e la fase circa -180°. Possiamo sempre continuare ad usare la funzione ***csim.***

```
-->w=0.3; //rad/sec;
-->t=0:0.1:100;
-->u=sin(w*t);
-->h=syslin('c',1/(s^2+0.5*s+1));
-->y=csim(u,t,h);
-->plot2d(t,u, color("red"))

-->plot2d(t,y, color("green"))
```



L'uscita (verde) segue abbastanza bene l'ingresso (rosso) con un piccolo ritardo.

Facciamo lo stesso variando la frequenza.

-->w=3; //rad/sec;

-->t=0:0.1:100;

-->u=sin(w*t);

-->h=syslin('c',1/(s^2+0.5*s+1));

-->y=csim(u,t,h);

-->plot2d(t,u, color("red"))

-->plot2d(t,y, color("green"))



Il modulo dell'uscita è circa 1/10 dell'ingresso e la fase è in ritardo di circa -180°. Ovviamente siamo già in una situazione di regime con il transitorio terminato.

Metodi di costruzione di un controllore

Supponiamo di avere un sistema oggetto dato da H(s) = 10/(1.25*s + 1), che viene messo in retroazione insieme ad un controllore che deve fornire:

1. errore nullo a regime;
2. massima elongazione possibile minore del 40%;
3. transitorio minore di 2 secondi.

Esistono 2 metodi per la risoluzione del problema: il **metodo grafico** ed il **metodo numerico**. Usando Scilab, ovviamente, il metodo più consono è quello grafico. Per prima cosa disegniamo il diagramma di Bode:

```
-->s=poly(0,'s');
-->N=10;
```

```
-->D=1.25*s +1;
-->h=N/D;
-->sl=syslin('c',h);
-->bode(sl)
```



Vediamo le caratteristiche che derivano direttamente da una attenta osservazione dei diagrammi.

Notiamo, inizialmente, che la B_P è di circa 0.16 Hz corrispondenti a 1 rad/sec. Il tempo di salita è dato approssimativamente al rapporto tra 1.8 e la pulsazione naturale ω_n che possiamo in questo caso approssimare alla banda passante. Pertanto, il tempo di salita, possiamo dire, è circa di 2 secondi.

Il margine di fase corrisponde a circa 95° alla frequenza di 1.27 Hz (8 rad/sec). Questo corrisponde ad un coefficiente di smorzamento di 0.95. Inserendo questo valore nell'equazione che lega la sovraelongazione con lo smorzamento, si trova che la sovraelongazione è circa 1%. Il sistema è eccessivamente smorzato.

Anche l'errore a regime può essere visto direttamente dal diagramma di Bode. Le costanti di posizione, di velocità o di accelerazione (K_p, K_v, K_a) si trovano nelle intersezioni dell'asintoto a bassa frequenza con la retta a $\omega = 1$. Il modulo in questi punti è costante. In questo caso l'asintoto è una linea orizzontale, con pendenza = 0; sappiamo, inoltre, che il sistema è di tipo 0.

Il guadagno è 20 dB e questo vuol dire che la costante della funzione errore è Kp = 10. In definitiva, l'errore a regime è $1/(1+K_p)=1/(1+10)=0.091$.

Tutto ciò che siamo riusciti a dedurre dal diagramma di Bode, è possibile vederlo anche con la risposta al gradino unitario, per adesso, senza controllore.

```
-->t=0:0.01:0.5;
-->R=1;
-->h1=h/.R;
-->sl=syslin('c',h1);
-->plot2d([t],[(csim('step',t,sl))])
```



Vediamo che il sistema è sovrasmorzato, ha un tempo di salita di circa 2 secondi, ed ha un errore a regime di circa il 9%.

Dobbiamo costruire il controllore che soddisfi le condizioni date. Dovendo avere un errore a regime nullo è d'obbligo scegliere un controllore PI. La generica f.d.T. è

$$G_C(s)=\frac{K*(s+a)}{s}$$

Il primo passo è trovare il coefficiente di smorzamento che corrisponde ad una sovraelongazione del 40%. Inserendo questo valore nell'equazione che lega lo smorzamento alla sovraelongazione, si ottiene un coefficiente pari a 0.28. Quindi, il nostro margine di fase è di circa 30°. ??????? Facendo i conti viene una B_P maggiore o uguale a 12 se vogliamo un tempo di salita minore di 1.75 secondi. Ricordiamo che stiamo analizzando il diagramma di Bode di un sistema ad anello aperto. Pertanto, la banda passante deve avere un guadagno di circa -7 dB.

Vediamo come influenza il PI la nostra risposta (supponiamo solo il termine integrale).

```
-->gs=1/s;
-->g=gs*h;
-->sl1=syslin('c',g);
-->bode(sl1)
```



Il margine di fase è di appena 16° e la banda passante è troppo piccola. Dobbiamo aggiungere con uno zero sia il modulo che la fase. Inseriamo uno zero in 1 e vediamo cosa succede.

```
-->gs2=(1+s)/s;
-->g2=gs2*h;
-->sl2=syslin('c',g2);
-->bode(sl2)
```



La risposta è soddisfacente. Il margine di fase è circa 88° e la B_P è circa 11 rad/sec. Cerchiamo di migliorare la risposta, aumentando la banda passante e mantenendo il margine di fase su quei valori. Incrementiamo, allora, il guadagno di 5:

```
-->gs3=(5+5*s)/s;

-->g3=gs3*h;

-->sl3=syslin('c',g3);

-->bode(sl3)

-->bode(sl3)
```



Il risultato è veramente ottimo, come si può vedere anche con la risposta allo step:

-->g3R=g3/.R;

-->ss=syslin('c',g3R);

-->plot2d([t],[(csim('step',t,ss))])